
NAG Library Routine Document

D03PHF/D03PHA

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

D03PHF/D03PHA integrates a system of linear or nonlinear parabolic partial differential equations
(PDEs) in one space variable, with scope for coupled ordinary differential equations (ODEs). The spatial
discretization is performed using finite differences, and the method of lines is employed to reduce the
PDEs to a system of ODEs. The resulting system is solved using a backward differentiation formula
method or a Theta method (switching between Newton’s method and functional iteration).

D03PHA is a version of D03PHF that has additional parameters in order to make it safe for use in
multithreaded applications (see Section 5).

2 Specification

2.1 Specification for D03PHF

SUBROUTINE D03PHF (NPDE, M, TS, TOUT, PDEDEF, BNDARY, U, NPTS, X, NCODE,
ODEDEF, NXI, XI, NEQN, RTOL, ATOL, ITOL, NORM, LAOPT,
ALGOPT, RSAVE, LRSAVE, ISAVE, LISAVE, ITASK, ITRACE,
IND, IFAIL)

&
&
&

INTEGER NPDE, M, NPTS, NCODE, NXI, NEQN, ITOL, LRSAVE,
ISAVE(LISAVE), LISAVE, ITASK, ITRACE, IND, IFAIL

&

REAL (KIND=nag_wp) TS, TOUT, U(NEQN), X(NPTS), XI(NXI), RTOL(*),
ATOL(*), ALGOPT(30), RSAVE(LRSAVE)

&

CHARACTER(1) NORM, LAOPT
EXTERNAL PDEDEF, BNDARY, ODEDEF

2.2 Specification for D03PHA

SUBROUTINE D03PHA (NPDE, M, TS, TOUT, PDEDEF, BNDARY, U, NPTS, X, NCODE,
ODEDEF, NXI, XI, NEQN, RTOL, ATOL, ITOL, NORM, LAOPT,
ALGOPT, RSAVE, LRSAVE, ISAVE, LISAVE, ITASK, ITRACE,
IND, IUSER, RUSER, CWSAV, LWSAV, IWSAV, RWSAV, IFAIL)

&
&
&

INTEGER NPDE, M, NPTS, NCODE, NXI, NEQN, ITOL, LRSAVE,
ISAVE(LISAVE), LISAVE, ITASK, ITRACE, IND,
IUSER(*), IWSAV(505), IFAIL

&
&

REAL (KIND=nag_wp) TS, TOUT, U(NEQN), X(NPTS), XI(NXI), RTOL(*),
ATOL(*), ALGOPT(30), RSAVE(LRSAVE), RUSER(*),
RWSAV(1100)

&
&

LOGICAL LWSAV(100)
CHARACTER(1) NORM, LAOPT
CHARACTER(80) CWSAV(10)
EXTERNAL PDEDEF, BNDARY, ODEDEF

3 Description

D03PHF/D03PHA integrates the system of parabolic-elliptic equations and coupled ODEs

XNPDE

j¼1

Pi;j
@Uj
@t
þQi ¼ x�m

@

@x
xmRið Þ; i ¼ 1; 2; . . . ;NPDE; a � x � b; t � t0; ð1Þ

Fi t; V ; _V ; �; U�; U�x; R
�; U�t ; U

�
xt

� �
¼ 0; i ¼ 1; 2; . . . ;NCODE; ð2Þ

where (1) defines the PDE part and (2) generalizes the coupled ODE part of the problem.

In (1), Pi;j and Ri depend on x, t, U , Ux and V ; Qi depends on x, t, U , Ux, V and linearly on _V . The
vector U is the set of PDE solution values

U x; tð Þ ¼ U1 x; tð Þ; . . . ; UNPDE x; tð Þ½ �T;

and the vector Ux is the partial derivative with respect to x. The vector V is the set of ODE solution
values

V tð Þ ¼ V1 tð Þ; . . . ; VNCODE tð Þ½ �T;

and _V denotes its derivative with respect to time.

In (2), � represents a vector of n� spatial coupling points at which the ODEs are coupled to the PDEs.
These points may or may not be equal to some of the PDE spatial mesh points. U�, U�x , R�, U�t and U�xt
are the functions U , Ux, R, Ut and Uxt evaluated at these coupling points. Each Fi may only depend
linearly on time derivatives. Hence the equation (2) may be written more precisely as

F ¼ G�A _V �B U�t
U�xt

� �
; ð3Þ

where F ¼ F1; . . . ; FNCODE½ �T, G is a vector of length NCODE, A is an NCODE by NCODE matrix, B
is an NCODE by n� � NPDE

� �
matrix and the entries in G, A and B may depend on t, �, U�, U�x and V .

In practice you only need to supply a vector of information to define the ODEs and not the matrices A
and B. (See Section 5 for the specification of ODEDEF.)

The integration in time is from t0 to tout, over the space interval a � x � b, where a ¼ x1 and b ¼ xNPTS

are the leftmost and rightmost points of a user-defined mesh x1; x2; . . . ; xNPTS. The coordinate system in
space is defined by the values of m; m ¼ 0 for Cartesian coordinates, m ¼ 1 for cylindrical polar
coordinates and m ¼ 2 for spherical polar coordinates.

The PDE system which is defined by the functions Pi;j, Qi and Ri must be specified in PDEDEF.

The initial values of the functions U x; tð Þ and V tð Þ must be given at t ¼ t0.

The functions Ri which may be thought of as fluxes, are also used in the definition of the boundary
conditions. The boundary conditions must have the form

�i x; tð ÞRi x; t; U; Ux; Vð Þ ¼ �i x; t; U; Ux; V ; _V
� �

; i ¼ 1; 2; . . . ;NPDE; ð4Þ

where x ¼ a or x ¼ b.

The boundary conditions must be specified in BNDARY. The function �i may depend linearly on _V .

The problem is subject to the following restrictions:

(i) In (1), _Vj tð Þ, for j ¼ 1; 2; . . . ;NCODE, may only appear linearly in the functions Qi, for
i ¼ 1; 2; . . . ;NPDE, with a similar restriction for �;

(ii) Pi;j and the flux Ri must not depend on any time derivatives;

(iii) t0 < tout, so that integration is in the forward direction;

(iv) the evaluation of the terms Pi;j, Qi and Ri is done approximately at the mid-points of the mesh
XðiÞ, for i ¼ 1; 2; . . . ;NPTS, by calling the PDEDEF for each mid-point in turn. Any discontinuities
in these functions must therefore be at one or more of the mesh points x1; x2; . . . ; xNPTS;

(v) at least one of the functions Pi;j must be nonzero so that there is a time derivative present in the
PDE problem;

(vi) if m > 0 and x1 ¼ 0:0, which is the left boundary point, then it must be ensured that the PDE
solution is bounded at this point. This can be done by either specifying the solution at x ¼ 0:0 or by
specifying a zero flux there, that is �i ¼ 1:0 and �i ¼ 0:0. See also Section 9 below.

The algebraic-differential equation system which is defined by the functions Fi must be specified in
ODEDEF. You must also specify the coupling points � in the array XI.

The parabolic equations are approximated by a system of ODEs in time for the values of Ui at mesh
points. For simple problems in Cartesian coordinates, this system is obtained by replacing the space
derivatives by the usual central, three-point finite difference formula. However, for polar and spherical
problems, or problems with nonlinear coefficients, the space derivatives are replaced by a modified three-
point formula which maintains second order accuracy. In total there are NPDE� NPTSþ NCODE ODEs
in the time direction. This system is then integrated forwards in time using a backward differentiation
formula (BDF) or a Theta method.

4 References

Berzins M (1990) Developments in the NAG Library software for parabolic equations Scientific Software
Systems (eds J C Mason and M G Cox) 59–72 Chapman and Hall

Berzins M, Dew P M and Furzeland R M (1989) Developing software for time-dependent problems
using the method of lines and differential-algebraic integrators Appl. Numer. Math. 5 375–397

Berzins M and Furzeland R M (1992) An adaptive theta method for the solution of stiff and nonstiff
differential equations Appl. Numer. Math. 9 1–19

Skeel R D and Berzins M (1990) A method for the spatial discretization of parabolic equations in one
space variable SIAM J. Sci. Statist. Comput. 11(1) 1–32

5 Parameters

1: NPDE – INTEGER Input

On entry: the number of PDEs to be solved.

Constraint: NPDE � 1.

2: M – INTEGER Input

On entry: the coordinate system used:

M ¼ 0
Indicates Cartesian coordinates.

M ¼ 1
Indicates cylindrical polar coordinates.

M ¼ 2
Indicates spherical polar coordinates.

Constraint: M ¼ 0, 1 or 2.

3: TS – REAL (KIND=nag_wp) Input/Output

On entry: the initial value of the independent variable t.

On exit: the value of t corresponding to the solution values in U. Normally TS ¼ TOUT.

Constraint: TS < TOUT.

4: TOUT – REAL (KIND=nag_wp) Input

On entry: the final value of t to which the integration is to be carried out.

5: PDEDEF – SUBROUTINE, supplied by the user. External Procedure

PDEDEF must evaluate the functions Pi;j, Qi and Ri which define the system of PDEs. The

functions may depend on x, t, U , Ux and V . Qi may depend linearly on _V . PDEDEF is called
approximately midway between each pair of mesh points in turn by D03PHF/D03PHA.

The specification of PDEDEF for D03PHF is:

SUBROUTINE PDEDEF (NPDE, T, X, U, UX, NCODE, V, VDOT, P, Q, R,
IRES)

&

INTEGER NPDE, NCODE, IRES
REAL (KIND=nag_wp) T, X, U(NPDE), UX(NPDE), V(NCODE),

VDOT(NCODE), P(NPDE,NPDE), Q(NPDE), R(NPDE)
&

The specification of PDEDEF for D03PHA is:

SUBROUTINE PDEDEF (NPDE, T, X, U, UX, NCODE, V, VDOT, P, Q, R,
IRES, IUSER, RUSER)

&

INTEGER NPDE, NCODE, IRES, IUSER(*)
REAL (KIND=nag_wp) T, X, U(NPDE), UX(NPDE), V(NCODE),

VDOT(NCODE), P(NPDE,NPDE), Q(NPDE), R(NPDE),
RUSER(*)

&
&

1: NPDE – INTEGER Input

On entry: the number of PDEs in the system.

2: T – REAL (KIND=nag_wp) Input

On entry: the current value of the independent variable t.

3: X – REAL (KIND=nag_wp) Input

On entry: the current value of the space variable x.

4: UðNPDEÞ – REAL (KIND=nag_wp) array Input

On entry: UðiÞ contains the value of the component Ui x; tð Þ, for i ¼ 1; 2; . . . ;NPDE.

5: UXðNPDEÞ – REAL (KIND=nag_wp) array Input

On entry: UXðiÞ contains the value of the component
@Ui x; tð Þ
@x

, for i ¼ 1; 2; . . . ;NPDE.

6: NCODE – INTEGER Input

On entry: the number of coupled ODEs in the system.

7: VðNCODEÞ – REAL (KIND=nag_wp) array Input

On entry: if NCODE > 0, VðiÞ contains the value of the component Vi tð Þ, for
i ¼ 1; 2; . . . ;NCODE.

8: VDOTðNCODEÞ – REAL (KIND=nag_wp) array Input

On entry: if NCODE > 0, VDOTðiÞ contains the value of component _Vi tð Þ, for
i ¼ 1; 2; . . . ;NCODE.

Note: _Vi tð Þ, for i ¼ 1; 2; . . . ;NCODE, may only appear linearly in Qj , for
j ¼ 1; 2; . . . ;NPDE.

9: PðNPDE;NPDEÞ – REAL (KIND=nag_wp) array Output

On exit: Pði; jÞ must be set to the value of Pi;j x; t; U; Ux; Vð Þ, for i ¼ 1; 2; . . . ;NPDE and
j ¼ 1; 2; . . . ;NPDE.

10: QðNPDEÞ – REAL (KIND=nag_wp) array Output

On exit: QðiÞ must be set to the value of Qi x; t; U; Ux; V ; _V
� �

, for i ¼ 1; 2; . . . ;NPDE.

D03PHF NAG Library Manual

D03PHF.4 Mark 25



11: RðNPDEÞ – REAL (KIND=nag_wp) array Output

On exit: RðiÞ must be set to the value of Ri x; t; U; Ux; Vð Þ, for i ¼ 1; 2; . . . ;NPDE.

12: IRES – INTEGER Input/Output

On entry: set to �1 or 1.

On exit: should usually remain unchanged. However, you may set IRES to force the
integration routine to take certain actions as described below:

IRES ¼ 2
Indicates to the integrator that control should be passed back immediately to the
calling (sub)routine with the error indicator set to IFAIL ¼ 6.

IRES ¼ 3
Indicates to the integrator that the current time step should be abandoned and a
smaller time step used instead. You may wish to set IRES ¼ 3 when a physically
meaningless input or output value has been generated. If you consecutively set
IRES ¼ 3, then D03PHF/D03PHA returns to the calling subroutine with the error
indicator set to IFAIL ¼ 4.

Note: the following are additional parameters for specific use with D03PHA. Users of D03PHF
therefore need not read the remainder of this description.

13: IUSERð�Þ – INTEGER array User Workspace
14: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

PDEDEF is called with the parameters IUSER and RUSER as supplied to D03PHF/
D03PHA. You are free to use the arrays IUSER and RUSER to supply information to
PDEDEF as an alternative to using COMMON global variables.

PDEDEF must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which D03PHF/D03PHA is called. Parameters denoted as Input must not be
changed by this procedure.

6: BNDARY – SUBROUTINE, supplied by the user. External Procedure

BNDARY must evaluate the functions �i and �i which describe the boundary conditions, as given
in (4).

The specification of BNDARY for D03PHF is:

SUBROUTINE BNDARY (NPDE, T, U, UX, NCODE, V, VDOT, IBND, BETA,
GAMMA, IRES)

&

INTEGER NPDE, NCODE, IBND, IRES
REAL (KIND=nag_wp) T, U(NPDE), UX(NPDE), V(NCODE), VDOT(NCODE),

BETA(NPDE), GAMMA(NPDE)
&

The specification of BNDARY for D03PHA is:

SUBROUTINE BNDARY (NPDE, T, U, UX, NCODE, V, VDOT, IBND, BETA,
GAMMA, IRES, IUSER, RUSER)

&

INTEGER NPDE, NCODE, IBND, IRES, IUSER(*)
REAL (KIND=nag_wp) T, U(NPDE), UX(NPDE), V(NCODE), VDOT(NCODE),

BETA(NPDE), GAMMA(NPDE), RUSER(*)
&

1: NPDE – INTEGER Input

On entry: the number of PDEs in the system.

2: T – REAL (KIND=nag_wp) Input

On entry: the current value of the independent variable t.

3: UðNPDEÞ – REAL (KIND=nag_wp) array Input

On entry: UðiÞ contains the value of the component Ui x; tð Þ at the boundary specified by
IBND, for i ¼ 1; 2; . . . ;NPDE.

4: UXðNPDEÞ – REAL (KIND=nag_wp) array Input

On entry: UXðiÞ contains the value of the component
@Ui x; tð Þ
@x

at the boundary specified

by IBND, for i ¼ 1; 2; . . . ;NPDE.

5: NCODE – INTEGER Input

On entry: the number of coupled ODEs in the system.

6: VðNCODEÞ – REAL (KIND=nag_wp) array Input

On entry: if NCODE > 0, VðiÞ contains the value of the component Vi tð Þ, for
i ¼ 1; 2; . . . ;NCODE.

7: VDOTðNCODEÞ – REAL (KIND=nag_wp) array Input

On entry: if NCODE > 0, VDOTðiÞ contains the value of component _Vi tð Þ, for
i ¼ 1; 2; . . . ;NCODE.

Note: _Vi tð Þ, for i ¼ 1; 2; . . . ;NCODE, may only appear linearly in Qj , for
j ¼ 1; 2; . . . ;NPDE.

8: IBND – INTEGER Input

On entry: specifies which boundary conditions are to be evaluated.

IBND ¼ 0
BNDARY must set up the coefficients of the left-hand boundary, x ¼ a.

IBND 6¼ 0
BNDARY must set up the coefficients of the right-hand boundary, x ¼ b.

9: BETAðNPDEÞ – REAL (KIND=nag_wp) array Output

On exit: BETAðiÞ must be set to the value of �i x; tð Þ at the boundary specified by IBND,
for i ¼ 1; 2; . . . ;NPDE.

10: GAMMAðNPDEÞ – REAL (KIND=nag_wp) array Output

On exit: GAMMAðiÞ must be set to the value of �i x; t; U; Ux; V ; _V
� �

at the boundary
specified by IBND, for i ¼ 1; 2; . . . ;NPDE.

11: IRES – INTEGER Input/Output

On entry: set to �1 or 1.

On exit: should usually remain unchanged. However, you may set IRES to force the
integration routine to take certain actions as described below:

IRES ¼ 2
Indicates to the integrator that control should be passed back immediately to the
calling (sub)routine with the error indicator set to IFAIL ¼ 6.

IRES ¼ 3
Indicates to the integrator that the current time step should be abandoned and a
smaller time step used instead. You may wish to set IRES ¼ 3 when a physically
meaningless input or output value has been generated. If you consecutively set
IRES ¼ 3, then D03PHF/D03PHA returns to the calling subroutine with the error
indicator set to IFAIL ¼ 4.

Note: the following are additional parameters for specific use with D03PHA. Users of D03PHF
therefore need not read the remainder of this description.

12: IUSERð�Þ – INTEGER array User Workspace
13: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

BNDARY is called with the parameters IUSER and RUSER as supplied to D03PHF/
D03PHA. You are free to use the arrays IUSER and RUSER to supply information to
BNDARY as an alternative to using COMMON global variables.

BNDARY must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which D03PHF/D03PHA is called. Parameters denoted as Input must not be
changed by this procedure.

7: UðNEQNÞ – REAL (KIND=nag_wp) array Input/Output

On entry: the initial values of the dependent variables defined as follows:

UðNPDE� j � 1ð Þ þ iÞ contain Ui xj ; t0
� �

, for i ¼ 1; 2; . . . ;NPDE and j ¼ 1; 2; . . . ;NPTS,
and

UðNPTS� NPDEþ iÞ contain Vi t0ð Þ, for i ¼ 1; 2; . . . ;NCODE.

On exit: the computed solution Ui xj ; t
� �

, for i ¼ 1; 2; . . . ;NPDE and j ¼ 1; 2; . . . ;NPTS, and
Vk tð Þ, for k ¼ 1; 2; . . . ;NCODE, evaluated at t ¼ TS.

8: NPTS – INTEGER Input

On entry: the number of mesh points in the interval a; b½ �.
Constraint: NPTS � 3.

9: XðNPTSÞ – REAL (KIND=nag_wp) array Input

On entry: the mesh points in the space direction. Xð1Þ must specify the left-hand boundary, a, and
XðNPTSÞ must specify the right-hand boundary, b.

Constraint: Xð1Þ < Xð2Þ < � � � < XðNPTSÞ.

10: NCODE – INTEGER Input

On entry: the number of coupled ODE components.

Constraint: NCODE � 0.

11: ODEDEF – SUBROUTINE, supplied by the NAG Library or the user. External Procedure

ODEDEF must evaluate the functions F , which define the system of ODEs, as given in (3).

If you wish to compute the solution of a system of PDEs only (NCODE ¼ 0), ODEDEF must be
the dummy routine D03PCK for D03PHF (or D53PCK for D03PHA). D03PCK and D53PCK are
included in the NAG Library.

The specification of ODEDEF for D03PHF is:

SUBROUTINE ODEDEF (NPDE, T, NCODE, V, VDOT, NXI, XI, UCP, UCPX,
RCP, UCPT, UCPTX, F, IRES)

&

INTEGER NPDE, NCODE, NXI, IRES
REAL (KIND=nag_wp) T, V(NCODE), VDOT(NCODE), XI(NXI),

UCP(NPDE,*), UCPX(NPDE,*), RCP(NPDE,*),
UCPT(NPDE,*), UCPTX(NPDE,*), F(NCODE)

&
&

The specification of ODEDEF for D03PHA is:

SUBROUTINE ODEDEF (NPDE, T, NCODE, V, VDOT, NXI, XI, UCP, UCPX,
RCP, UCPT, UCPTX, F, IRES, IUSER, RUSER)

&

INTEGER NPDE, NCODE, NXI, IRES, IUSER(*)
REAL (KIND=nag_wp) T, V(NCODE), VDOT(NCODE), XI(NXI),

UCP(NPDE,*), UCPX(NPDE,*), RCP(NPDE,*),
UCPT(NPDE,*), UCPTX(NPDE,*), F(NCODE),
RUSER(*)

&
&
&

1: NPDE – INTEGER Input

On entry: the number of PDEs in the system.

2: T – REAL (KIND=nag_wp) Input

On entry: the current value of the independent variable t.

3: NCODE – INTEGER Input

On entry: the number of coupled ODEs in the system.

4: VðNCODEÞ – REAL (KIND=nag_wp) array Input

On entry: if NCODE > 0, VðiÞ contains the value of the component Vi tð Þ, for
i ¼ 1; 2; . . . ;NCODE.

5: VDOTðNCODEÞ – REAL (KIND=nag_wp) array Input

On entry: if NCODE > 0, VDOTðiÞ contains the value of component _Vi tð Þ, for
i ¼ 1; 2; . . . ;NCODE.

6: NXI – INTEGER Input

On entry: the number of ODE/PDE coupling points.

7: XIðNXIÞ – REAL (KIND=nag_wp) array Input

On entry: if NXI > 0, XIðiÞ contains the ODE/PDE coupling points, �i, for
i ¼ 1; 2; . . . ;NXI.

8: UCPðNPDE; �Þ – REAL (KIND=nag_wp) array Input

On entry: if NXI > 0, UCPði; jÞ contains the value of Ui x; tð Þ at the coupling point
x ¼ �j , for i ¼ 1; 2; . . . ;NPDE and j ¼ 1; 2; . . . ;NXI.

9: UCPXðNPDE; �Þ – REAL (KIND=nag_wp) array Input

On entry: if NXI > 0, UCPXði; jÞ contains the value of
@Ui x; tð Þ
@x

at the coupling point

x ¼ �j , for i ¼ 1; 2; . . . ;NPDE and j ¼ 1; 2; . . . ;NXI.

10: RCPðNPDE; �Þ – REAL (KIND=nag_wp) array Input

On entry: RCPði; jÞ contains the value of the flux Ri at the coupling point x ¼ �j , for
i ¼ 1; 2; . . . ;NPDE and j ¼ 1; 2; . . . ;NXI.

11: UCPTðNPDE; �Þ – REAL (KIND=nag_wp) array Input

On entry: if NXI > 0, UCPTði; jÞ contains the value of
@Ui

@t
at the coupling point x ¼ �j ,

for i ¼ 1; 2; . . . ;NPDE and j ¼ 1; 2; . . . ;NXI.

D03PHF NAG Library Manual

D03PHF.8 Mark 25



12: UCPTXðNPDE; �Þ – REAL (KIND=nag_wp) array Input

On entry: UCPTXði; jÞ contains the value of
@2Ui

@x@t
at the coupling point x ¼ �j , for

i ¼ 1; 2; . . . ;NPDE and j ¼ 1; 2; . . . ;NXI.

13: FðNCODEÞ – REAL (KIND=nag_wp) array Output

On exit: FðiÞ must contain the ith component of F , for i ¼ 1; 2; . . . ;NCODE, where F is
defined as

F ¼ G�A _V �B U�t
U�xt

� �
; ð5Þ

or

F ¼ �A _V �B U�t
U�xt

� �
: ð6Þ

The definition of F is determined by the input value of IRES.

14: IRES – INTEGER Input/Output

On entry: the form of F that must be returned in the array F.

IRES ¼ 1
Equation (5) must be used.

IRES ¼ �1
Equation (6) must be used.

On exit: should usually remain unchanged. However, you may reset IRES to force the
integration routine to take certain actions as described below:

IRES ¼ 2
Indicates to the integrator that control should be passed back immediately to the
calling (sub)routine with the error indicator set to IFAIL ¼ 6.

IRES ¼ 3
Indicates to the integrator that the current time step should be abandoned and a
smaller time step used instead. You may wish to set IRES ¼ 3 when a physically
meaningless input or output value has been generated. If you consecutively set
IRES ¼ 3, then D03PHF/D03PHA returns to the calling subroutine with the error
indicator set to IFAIL ¼ 4.

Note: the following are additional parameters for specific use with D03PHA. Users of D03PHF
therefore need not read the remainder of this description.

15: IUSERð�Þ – INTEGER array User Workspace
16: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

ODEDEF is called with the parameters IUSER and RUSER as supplied to D03PHF/
D03PHA. You are free to use the arrays IUSER and RUSER to supply information to
ODEDEF as an alternative to using COMMON global variables.

ODEDEF must either be a module subprogram USEd by, or declared as EXTERNAL in, the
(sub)program from which D03PHF/D03PHA is called. Parameters denoted as Input must not be
changed by this procedure.

12: NXI – INTEGER Input

On entry: the number of ODE/PDE coupling points.

Constraints:

if NCODE ¼ 0, NXI ¼ 0;
if NCODE > 0, NXI � 0.

13: XIðNXIÞ – REAL (KIND=nag_wp) array Input

On entry: if NXI > 0, XIðiÞ, for i ¼ 1; 2; . . . ;NXI, must be set to the ODE/PDE coupling points.

Constraint: Xð1Þ � XIð1Þ < XIð2Þ < � � � < XIðNXIÞ � XðNPTSÞ.

14: NEQN – INTEGER Input

On entry: the number of ODEs in the time direction.

Constraint: NEQN ¼ NPDE� NPTSþ NCODE.

15: RTOLð�Þ – REAL (KIND=nag_wp) array Input

Note: the dimension of the array RTOL must be at least 1 if ITOL ¼ 1 or 2 and at least NEQN if
ITOL ¼ 3 or 4.

On entry: the relative local error tolerance.

Constraint: RTOLðiÞ � 0:0 for all relevant i.

16: ATOLð�Þ – REAL (KIND=nag_wp) array Input

Note: the dimension of the array ATOL must be at least 1 if ITOL ¼ 1 or 3 and at least NEQN if
ITOL ¼ 2 or 4.

On entry: the absolute local error tolerance.

Constraint: ATOLðiÞ � 0:0 for all relevant i.

Note: corresponding elements of RTOL and ATOL cannot both be 0:0.

17: ITOL – INTEGER Input

On entry: a value to indicate the form of the local error test. ITOL indicates to D03PHF/D03PHA
whether to interpret either or both of RTOL or ATOL as a vector or scalar. The error test to be
satisfied is ei=wik k < 1:0, where wi is defined as follows:

ITOL RTOL ATOL wi
1 scalar scalar RTOLð1Þ � Uij j þ ATOLð1Þ
2 scalar vector RTOLð1Þ � Uij j þ ATOLðiÞ
3 vector scalar RTOLðiÞ � Uij j þ ATOLð1Þ
4 vector vector RTOLðiÞ � Uij j þ ATOLðiÞ

In the above, ei denotes the estimated local error for the ith component of the coupled PDE/ODE
system in time, UðiÞ, for i ¼ 1; 2; . . . ;NEQN.

The choice of norm used is defined by the parameter NORM.

Constraint: 1 � ITOL � 4.

18: NORM – CHARACTER(1) Input

On entry: the type of norm to be used.

NORM ¼ M
Maximum norm.

NORM ¼ A
Averaged L2 norm.

If Unorm denotes the norm of the vector U of length NEQN, then for the averaged L2 norm

Unorm ¼

ffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffiffi
1

NEQN

XNEQN

i¼1

UðiÞ=wið Þ2
vuut ;

while for the maximum norm

Unorm ¼ max
i

UðiÞ=wij j:

See the description of ITOL for the formulation of the weight vector w.

Constraint: NORM ¼ M or A .

19: LAOPT – CHARACTER(1) Input

On entry: the type of matrix algebra required.

LAOPT ¼ F
Full matrix methods to be used.

LAOPT ¼ B
Banded matrix methods to be used.

LAOPT ¼ S
Sparse matrix methods to be used.

Constraint: LAOPT ¼ F , B or S .

Note: you are recommended to use the banded option when no coupled ODEs are present (i.e.,
NCODE ¼ 0).

20: ALGOPTð30Þ – REAL (KIND=nag_wp) array Input

On entry: may be set to control various options available in the integrator. If you wish to employ
all the default options, then ALGOPTð1Þ should be set to 0:0. Default values will also be used for
any other elements of ALGOPT set to zero. The permissible values, default values, and meanings
are as follows:

ALGOPTð1Þ
Selects the ODE integration method to be used. If ALGOPTð1Þ ¼ 1:0, a BDF method is
used and if ALGOPTð1Þ ¼ 2:0, a Theta method is used. The default value is
ALGOPTð1Þ ¼ 1:0.

If ALGOPTð1Þ ¼ 2:0, then ALGOPTðiÞ, for i ¼ 2; 3; 4 are not used.

ALGOPTð2Þ
Specifies the maximum order of the BDF integration formula to be used. ALGOPTð2Þ may
be 1:0, 2:0, 3:0, 4:0 or 5:0. The default value is ALGOPTð2Þ ¼ 5:0.

ALGOPTð3Þ
Specifies what method is to be used to solve the system of nonlinear equations arising on
each step of the BDF method. If ALGOPTð3Þ ¼ 1:0 a modified Newton iteration is used
and if ALGOPTð3Þ ¼ 2:0 a functional iteration method is used. If functional iteration is
selected and the integrator encounters difficulty, then there is an automatic switch to the
modified Newton iteration. The default value is ALGOPTð3Þ ¼ 1:0.

ALGOPTð4Þ
Specifies whether or not the Petzold error test is to be employed. The Petzold error test
results in extra overhead but is more suitable when algebraic equations are present, such as

Pi;j ¼ 0:0, for j ¼ 1; 2; . . . ;NPDE, for some i or when there is no _Vi tð Þ dependence in the
coupled ODE system. If ALGOPTð4Þ ¼ 1:0, then the Petzold test is used. If
ALGOPTð4Þ ¼ 2:0, then the Petzold test is not used. The default value is
ALGOPTð4Þ ¼ 1:0.

If ALGOPTð1Þ ¼ 1:0, then ALGOPTðiÞ, for i ¼ 5; 6; 7, are not used.

ALGOPTð5Þ
Specifies the value of Theta to be used in the Theta integration method.
0:51 � ALGOPTð5Þ � 0:99. The default value is ALGOPTð5Þ ¼ 0:55.

ALGOPTð6Þ
Specifies what method is to be used to solve the system of nonlinear equations arising on
each step of the Theta method. If ALGOPTð6Þ ¼ 1:0, a modified Newton iteration is used
and if ALGOPTð6Þ ¼ 2:0, a functional iteration method is used. The default value is
ALGOPTð6Þ ¼ 1:0.

ALGOPTð7Þ
Specifies whether or not the integrator is allowed to switch automatically between modified
Newton and functional iteration methods in order to be more efficient. If
ALGOPTð7Þ ¼ 1:0, then switching is allowed and if ALGOPTð7Þ ¼ 2:0, then switching
is not allowed. The default value is ALGOPTð7Þ ¼ 1:0.

ALGOPTð11Þ
Specifies a point in the time direction, tcrit, beyond which integration must not be
attempted. The use of tcrit is described under the parameter ITASK. If ALGOPTð1Þ 6¼ 0:0, a
value of 0:0 for ALGOPTð11Þ, say, should be specified even if ITASK subsequently
specifies that tcrit will not be used.

ALGOPTð12Þ
Specifies the minimum absolute step size to be allowed in the time integration. If this
option is not required, ALGOPTð12Þ should be set to 0:0.

ALGOPTð13Þ
Specifies the maximum absolute step size to be allowed in the time integration. If this
option is not required, ALGOPTð13Þ should be set to 0:0.

ALGOPTð14Þ
Specifies the initial step size to be attempted by the integrator. If ALGOPTð14Þ ¼ 0:0, then
the initial step size is calculated internally.

ALGOPTð15Þ
Specifies the maximum number of steps to be attempted by the integrator in any one call. If
ALGOPTð15Þ ¼ 0:0, then no limit is imposed.

ALGOPTð23Þ
Specifies what method is to be used to solve the nonlinear equations at the initial point to

initialize the values of U , Ut, V and _V . If ALGOPTð23Þ ¼ 1:0, a modified Newton iteration
is used and if ALGOPTð23Þ ¼ 2:0, functional iteration is used. The default value is
ALGOPTð23Þ ¼ 1:0.

ALGOPTð29Þ and ALGOPTð30Þ are used only for the sparse matrix algebra option, LAOPT ¼ S .

ALGOPTð29Þ
Governs the choice of pivots during the decomposition of the first Jacobian matrix. It
should lie in the range 0:0 < ALGOPTð29Þ < 1:0, with smaller values biasing the algorithm
towards maintaining sparsity at the expense of numerical stability. If ALGOPTð29Þ lies
outside this range then the default value is used. If the routines regard the Jacobian matrix
as numerically singular then increasing ALGOPTð29Þ towards 1:0 may help, but at the cost
of increased fill-in. The default value is ALGOPTð29Þ ¼ 0:1.

ALGOPTð30Þ
Is used as a relative pivot threshold during subsequent Jacobian decompositions (see
ALGOPTð29Þ) below which an internal error is invoked. If ALGOPTð30Þ is greater than
1:0 no check is made on the pivot size, and this may be a necessary option if the Jacobian
is found to be numerically singular (see ALGOPTð29Þ). The default value is
ALGOPTð30Þ ¼ 0:0001.

21: RSAVEðLRSAVEÞ – REAL (KIND=nag_wp) array Communication Array

If IND ¼ 0, RSAVE need not be set on entry.

If IND ¼ 1, RSAVE must be unchanged from the previous call to the routine because it contains
required information about the iteration.

22: LRSAVE – INTEGER Input

On entry: the dimension of the array RSAVE as declared in the (sub)program from which
D03PHF/D03PHA is called.

Constraint:

If LAOPT ¼ F , LRSAVE � NEQN� NEQNþ NEQNþ nwkres þ lenode.

If LAOPT ¼ B , LRSAVE � 3�mlu þ 1ð Þ � NEQNþ nwkres þ lenode.

If LAOPT ¼ S , LRSAVE � 4� NEQNþ 11� NEQN=2þ 1þ nwkres þ lenode.

Where

mlu is the lower or upper half bandwidths such that
mlu ¼ 3� NPDE� 1, for PDE problems only (no coupled ODEs); or
mlu ¼ NEQN� 1, for coupled PDE/ODE problems.

nwkres ¼
NPDE� 2� NPTSþ 6� NXIþ 3� NPDEþ 26ð Þ þ NXIþ NCODEþ 7� NPTSþ 2; when NCODE > 0 and NXI > 0;
NPDE� 2� NPTSþ 3� NPDEþ 32ð Þ þ NCODEþ 7� NPTSþ 3; when NCODE > 0 and NXI ¼ 0;
NPDE� 2� NPTSþ 3� NPDEþ 32ð Þ þ 7� NPTSþ 4; when NCODE ¼ 0:

8<
:

lenode ¼ 6þ int ALGOPTð2Þð Þð Þ � NEQNþ 50; when the BDF method is used; or
9� NEQNþ 50; when the Theta method is used:

�

Note: when LAOPT ¼ S , the value of LRSAVE may be too small when supplied to the
integrator. An estimate of the minimum size of LRSAVE is printed on the current error message
unit if ITRACE > 0 and the routine returns with IFAIL ¼ 15..

23: ISAVEðLISAVEÞ – INTEGER array Communication Array

If IND ¼ 0, ISAVE need not be set on entry.

If IND ¼ 1, ISAVE must be unchanged from the previous call to the routine because it contains
required information about the iteration. In particular:

ISAVEð1Þ
Contains the number of steps taken in time.

ISAVEð2Þ
Contains the number of residual evaluations of the resulting ODE system used. One such
evaluation involves computing the PDE functions at all the mesh points, as well as one
evaluation of the functions in the boundary conditions.

ISAVEð3Þ
Contains the number of Jacobian evaluations performed by the time integrator.

ISAVEð4Þ
Contains the order of the last backward differentiation formula method used.

ISAVEð5Þ
Contains the number of Newton iterations performed by the time integrator. Each iteration
involves an ODE residual evaluation followed by a back-substitution using the LU
decomposition of the Jacobian matrix.

24: LISAVE – INTEGER Input

On entry: the dimension of the array ISAVE as declared in the (sub)program from which D03PHF/
D03PHA is called. Its size depends on the type of matrix algebra selected:

if LAOPT ¼ F , LISAVE � 24;

if LAOPT ¼ B , LISAVE � NEQNþ 24;

if LAOPT ¼ S , LISAVE � 25� NEQNþ 24.

Note: when using the sparse option, the value of LISAVE may be too small when supplied to the
integrator. An estimate of the minimum size of LISAVE is printed on the current error message
unit if ITRACE > 0 and the routine returns with IFAIL ¼ 15.

25: ITASK – INTEGER Input

On entry: specifies the task to be performed by the ODE integrator.

ITASK ¼ 1
Normal computation of output values U at t ¼ TOUT.

ITASK ¼ 2
One step and return.

ITASK ¼ 3
Stop at first internal integration point at or beyond t ¼ TOUT.

ITASK ¼ 4
Normal computation of output values U at t ¼ TOUT but without overshooting t ¼ tcrit

where tcrit is described under the parameter ALGOPT.

ITASK ¼ 5
Take one step in the time direction and return, without passing tcrit, where tcrit is described
under the parameter ALGOPT.

Constraint: ITASK ¼ 1, 2, 3, 4 or 5.

26: ITRACE – INTEGER Input

On entry: the level of trace information required from D03PHF/D03PHA and the underlying ODE
solver. ITRACE may take the value �1, 0, 1, 2 or 3.

ITRACE ¼ �1
No output is generated.

ITRACE ¼ 0
Only warning messages from the PDE solver are printed on the current error message unit
(see X04AAF).

ITRACE > 0
Output from the underlying ODE solver is printed on the current advisory message unit (see
X04ABF). This output contains details of Jacobian entries, the nonlinear iteration and the
time integration during the computation of the ODE system.

If ITRACE < �1, then �1 is assumed and similarly if ITRACE > 3, then 3 is assumed.

The advisory messages are given in greater detail as ITRACE increases. You are advised to set
ITRACE ¼ 0, unless you are experienced with sub-chapter D02M–N.

27: IND – INTEGER Input/Output

On entry: indicates whether this is a continuation call or a new integration.

IND ¼ 0
Starts or restarts the integration in time.

IND ¼ 1
Continues the integration after an earlier exit from the routine. In this case, only the
parameters TOUT and IFAIL should be reset between calls to D03PHF/D03PHA.

Constraint: IND ¼ 0 or 1.

On exit: IND ¼ 1.

28: IFAIL – INTEGER Input/Output

Note: for D03PHA, IFAIL does not occur in this position in the parameter list. See the additional
parameters described below.

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this parameter you should
refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

Note: the following are additional parameters for specific use with D03PHA. Users of D03PHF therefore
need not read the remainder of this description.

29: IUSERð�Þ – INTEGER array User Workspace
30: RUSERð�Þ – REAL (KIND=nag_wp) array User Workspace

IUSER and RUSER are not used by D03PHF/D03PHA, but are passed directly to PDEDEF,
BNDARY and ODEDEF and may be used to pass information to these routines as an alternative
to using COMMON global variables.

31: CWSAVð10Þ – CHARACTER(80) array Communication Array

32: LWSAVð100Þ – LOGICAL array Communication Array

33: IWSAVð505Þ – INTEGER array Communication Array

34: RWSAVð1100Þ – REAL (KIND=nag_wp) array Communication Array

35: IFAIL – INTEGER Input/Output

Note: see the parameter description for IFAIL above.

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit (as
defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, TOUT� TS is too small,
or ITASK 6¼ 1, 2, 3, 4 or 5,
or M 6¼ 0, 1 or 2,
or at least one of the coupling points defined in array XI is outside the interval

[Xð1Þ;XðNPTSÞ],
or M > 0 and Xð1Þ < 0:0,
or NPTS < 3,
or NPDE < 1,
or NORM 6¼ A or M ,
or LAOPT 6¼ F , B or S ,
or ITOL 6¼ 1, 2, 3 or 4,
or IND 6¼ 0 or 1,
or mesh points XðiÞ are badly ordered,
or LRSAVE is too small,

D03 – Partial Differential Equations D03PHF

Mark 25 D03PHF.15



or LISAVE is too small,
or NCODE and NXI are incorrectly defined,
or NEQN 6¼ NPDE� NPTSþ NCODE,
or either an element of RTOL or ATOL < 0:0,
or all the elements of RTOL and ATOL are zero.

IFAIL ¼ 2

The underlying ODE solver cannot make any further progress, with the values of ATOL and
RTOL, across the integration range from the current point t ¼ TS. The components of U contain
the computed values at the current point t ¼ TS.

IFAIL ¼ 3

In the underlying ODE solver, there were repeated error test failures on an attempted step, before
completing the requested task, but the integration was successful as far as t ¼ TS. The problem
may have a singularity, or the error requirement may be inappropriate.

IFAIL ¼ 4

In setting up the ODE system, the internal initialization routine was unable to initialize the
derivative of the ODE system. This could be due to the fact that IRES was repeatedly set to 3 in
at least PDEDEF, BNDARY or ODEDEF, when the residual in the underlying ODE solver was
being evaluated.

IFAIL ¼ 5

In solving the ODE system, a singular Jacobian has been encountered. You should check your
problem formulation.

IFAIL ¼ 6

When evaluating the residual in solving the ODE system, IRES was set to 2 in at least PDEDEF,
BNDARY or ODEDEF. Integration was successful as far as t ¼ TS.

IFAIL ¼ 7

The values of ATOL and RTOL are so small that the routine is unable to start the integration in
time.

IFAIL ¼ 8

In one of PDEDEF, BNDARY or ODEDEF, IRES was set to an invalid value.

IFAIL ¼ 9 (D02NNF)

A serious error has occurred in an internal call to the specified routine. Check the problem
specification and all parameters and array dimensions. Setting ITRACE ¼ 1 may provide more
information. If the problem persists, contact NAG.

IFAIL ¼ 10

The required task has been completed, but it is estimated that a small change in ATOL and RTOL
is unlikely to produce any change in the computed solution. (Only applies when you are not
operating in one step mode, that is when ITASK 6¼ 2 or 5.)

IFAIL ¼ 11

An error occurred during Jacobian formulation of the ODE system (a more detailed error
description may be directed to the current error message unit). If using the sparse matrix algebra
option, the values of ALGOPTð29Þ and ALGOPTð30Þ may be inappropriate.

IFAIL ¼ 12

In solving the ODE system, the maximum number of steps specified in ALGOPTð15Þ have been
taken.

IFAIL ¼ 13

Some error weights wi became zero during the time integration (see the description of ITOL).
Pure relative error control (ATOLðiÞ ¼ 0:0) was requested on a variable (the ith) which has
become zero. The integration was successful as far as t ¼ TS.

IFAIL ¼ 14

The flux function Ri was detected as depending on time derivatives, which is not permissible.

IFAIL ¼ 15

When using the sparse option, the value of LISAVE or LRSAVE was not sufficient (more detailed
information may be directed to the current error message unit).

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

D03PHF/D03PHA controls the accuracy of the integration in the time direction but not the accuracy of
the approximation in space. The spatial accuracy depends on both the number of mesh points and on
their distribution in space. In the time integration only the local error over a single step is controlled and
so the accuracy over a number of steps cannot be guaranteed. You should therefore test the effect of
varying the accuracy parameters ATOL and RTOL.

8 Parallelism and Performance

D03PHF/D03PHA is threaded by NAG for parallel execution in multithreaded implementations of the
NAG Library.

D03PHF/D03PHA makes calls to BLAS and/or LAPACK routines, which may be threaded within the
vendor library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users’ Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The parameter specification allows you to include equations with only first-order derivatives in the space
direction but there is no guarantee that the method of integration will be satisfactory for such systems.
The position and nature of the boundary conditions in particular are critical in defining a stable problem.

It may be advisable in such cases to reduce the whole system to first-order and to use the Keller box
scheme routine D03PKF.

The time taken depends on the complexity of the parabolic system and on the accuracy requested. For a
given system and a fixed accuracy it is approximately proportional to NEQN.

10 Example

This example provides a simple coupled system of one PDE and one ODE.

V1ð Þ2
@U1

@t
� xV1

_V1
@U1

@x
¼ @2yU1

@x2

_V1 ¼ V1U1 þ
@U1

@x
þ 1þ t;

for t 2 10�4; 0:1� 2i
� 	

; i ¼ 1; 2; . . . ; 5; x 2 0; 1½ �.

The left boundary condition at x ¼ 0 is

@U1

@x
¼ �V1 exp t:

The right boundary condition at x ¼ 1 is

@U1

@x
¼ �V1

_V1:

The initial conditions at t ¼ 10�4 are defined by the exact solution:

V1 ¼ t; and U1 x; tð Þ ¼ exp t 1� xð Þf g � 1:0; x 2 0; 1½ �;

and the coupling point is at �1 ¼ 1:0.

10.1 Program Text

the following program illustrates the use of D03PHF. An equivalent program illustrating the use of
D03PHA is available with the supplied Library and is also available from the NAG web site.

! D03PHF Example Program Text
! Mark 25 Release. NAG Copyright 2014.

Module d03phfe_mod

! D03PHF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: bndary, exact, odedef, pdedef, &

uvinit
! .. Parameters ..

Real (Kind=nag_wp), Parameter :: one = 1.0_nag_wp
Integer, Parameter, Public :: itrace = 0, ncode = 1, nin = 5, &

nout = 6, npde = 1, nxi = 1
! .. Local Scalars ..

Real (Kind=nag_wp), Public, Save :: ts
Contains

Subroutine odedef(npde,t,ncode,v,vdot,nxi,xi,ucp,ucpx,rcp,ucpt,ucptx,f, &
ires)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: t
Integer, Intent (Inout) :: ires
Integer, Intent (In) :: ncode, npde, nxi

! .. Array Arguments ..

Real (Kind=nag_wp), Intent (Out) :: f(ncode)
Real (Kind=nag_wp), Intent (In) :: rcp(npde,*), ucp(npde,*), &

ucpt(npde,*), ucptx(npde,*), &
ucpx(npde,*), v(ncode), &
vdot(ncode), xi(nxi)

! .. Executable Statements ..
If (ires==1) Then

f(1) = vdot(1) - v(1)*ucp(1,1) - ucpx(1,1) - one - t
Else If (ires==-1) Then

f(1) = vdot(1)
End If
Return

End Subroutine odedef
Subroutine pdedef(npde,t,x,u,ux,ncode,v,vdot,p,q,r,ires)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: t, x
Integer, Intent (Inout) :: ires
Integer, Intent (In) :: ncode, npde

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: p(npde,npde), q(npde), r(npde)
Real (Kind=nag_wp), Intent (In) :: u(npde), ux(npde), v(ncode), &

vdot(ncode)
! .. Executable Statements ..

p(1,1) = v(1)*v(1)
r(1) = ux(1)
q(1) = -x*ux(1)*v(1)*vdot(1)
Return

End Subroutine pdedef
Subroutine bndary(npde,t,u,ux,ncode,v,vdot,ibnd,beta,gamma,ires)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: t
Integer, Intent (In) :: ibnd, ncode, npde
Integer, Intent (Inout) :: ires

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: beta(npde), gamma(npde)
Real (Kind=nag_wp), Intent (In) :: u(npde), ux(npde), v(ncode), &

vdot(ncode)
! .. Intrinsic Procedures ..

Intrinsic :: exp
! .. Executable Statements ..

beta(1) = one
If (ibnd==0) Then

gamma(1) = -v(1)*exp(t)
Else

gamma(1) = -v(1)*vdot(1)
End If
Return

End Subroutine bndary
Subroutine uvinit(npde,npts,x,u,ncode,neqn)

! Routine for PDE initial values

! .. Scalar Arguments ..
Integer, Intent (In) :: ncode, neqn, npde, npts

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: u(neqn)
Real (Kind=nag_wp), Intent (In) :: x(npts)

! .. Local Scalars ..
Integer :: i

! .. Intrinsic Procedures ..
Intrinsic :: exp

! .. Executable Statements ..
Do i = 1, npts

u(i) = exp(ts*(one-x(i))) - one
End Do
u(neqn) = ts
Return

End Subroutine uvinit
Subroutine exact(time,npts,x,u)

! Exact solution (for comparison purpose)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: time
Integer, Intent (In) :: npts

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: u(npts)
Real (Kind=nag_wp), Intent (In) :: x(npts)

! .. Local Scalars ..
Integer :: i

! .. Intrinsic Procedures ..
Intrinsic :: exp

! .. Executable Statements ..
Do i = 1, npts

u(i) = exp(time*(one-x(i))) - one
End Do
Return

End Subroutine exact
End Module d03phfe_mod
Program d03phfe

! D03PHF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d03phf, nag_wp
Use d03phfe_mod, Only: bndary, exact, itrace, ncode, nin, nout, npde, &

nxi, odedef, pdedef, ts, uvinit
! .. Implicit None Statement ..

Implicit None
! .. Local Scalars ..

Real (Kind=nag_wp) :: tout
Integer :: i, ifail, ind, it, itask, itol, &

latol, lenode, lisave, lrsave, &
lrtol, m, neqn, npts, nwkres

Logical :: theta
Character (1) :: laopt, norm

! .. Local Arrays ..
Real (Kind=nag_wp) :: algopt(30), xi(nxi)
Real (Kind=nag_wp), Allocatable :: atol(:), exy(:), rsave(:), &

rtol(:), u(:), x(:)
Integer, Allocatable :: isave(:)

! .. Intrinsic Procedures ..
Intrinsic :: mod, real

! .. Executable Statements ..
Write (nout,*) ’D03PHF Example Program Results’

! Skip heading in data file
Read (nin,*)
Read (nin,*) m, npts
neqn = npde*npts + ncode
nwkres = npde*(npts+6*nxi+3*npde+15) + ncode + nxi + 7*npts + 2
lenode = 11*neqn + 50
lrsave = neqn*neqn + neqn + nwkres + lenode
lisave = 25*neqn + 24

Allocate (exy(npts),u(neqn),rsave(lrsave),x(npts),isave(lisave))

Read (nin,*) itol
latol = 1
lrtol = 1
If (itol>2) latol = neqn
If (mod(itol,2)==0) lrtol = neqn
Allocate (atol(latol),rtol(lrtol))
Read (nin,*) atol(1:latol), rtol(1:lrtol)

Read (nin,*) ts

! Set break-points
Do i = 1, npts

x(i) = real(i-1,kind=nag_wp)/real(npts-1,kind=nag_wp)
End Do

Read (nin,*) xi(1:nxi)
Read (nin,*) norm, laopt
ind = 0
itask = 1

! Set theta to .TRUE. if the Theta integrator is required
theta = .False.
algopt(1:30) = 0.0_nag_wp
If (theta) Then

algopt(1) = 2.0_nag_wp
End If

! Loop over output value of t

Call uvinit(npde,npts,x,u,ncode,neqn)
tout = 0.2_nag_wp
Do it = 1, 5

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call d03phf(npde,m,ts,tout,pdedef,bndary,u,npts,x,ncode,odedef,nxi,xi, &

neqn,rtol,atol,itol,norm,laopt,algopt,rsave,lrsave,isave,lisave, &
itask,itrace,ind,ifail)

If (it==1) Then
Write (nout,99997) atol, npts
Write (nout,99999) x(1:npts-5:4), x(npts)

End If

! Print against the exact solution.
Call exact(tout,npts,x,exy)
Write (nout,99998) ts
Write (nout,99995) u(1:npts-5:4), u(npts:neqn)
Write (nout,99994) exy(1:npts-5:4), exy(npts), ts

! Select next time to solve to for output.
tout = 2.0_nag_wp*tout

End Do
Write (nout,99996) isave(1), isave(2), isave(3), isave(5)

99999 Format (’ X ’,5F9.3/)
99998 Format (’ T = ’,F6.3)
99997 Format (//’ Simple coupled PDE using BDF ’/’ Accuracy require’, &

’ment =’,E10.3,’ Number of points = ’,I4/)
99996 Format (’ Number of integration steps in time = ’,I6/’ Number o’, &

’f function evaluations = ’,I6/’ Number of Jacobian eval’,’uations =’, &
I6/’ Number of iterations = ’,I6)

99995 Format (1X,’App. sol. ’,F7.3,4F9.3,’ ODE sol. =’,F8.3)
99994 Format (1X,’Exact sol. ’,F7.3,4F9.3,’ ODE sol. =’,F8.3/)

End Program d03phfe

10.2 Program Data

D03PHF Example Program Data
0 21 : m, npts
1 : itol (latol=1,lrtol=1)
1.0E-4, 1.0E-4 : atol(1), rtol(1)
1.0E-4 : ts
1.0 : xi(1:nxi)
A F : norm, laopt

10.3 Program Results

D03PHF Example Program Results

Simple coupled PDE using BDF
Accuracy requirement = 0.100E-03 Number of points = 21

X 0.000 0.200 0.400 0.600 1.000

D03 – Partial Differential Equations D03PHF

Mark 25 D03PHF.21



T = 0.200
App. sol. 0.222 0.174 0.128 0.084 0.001 ODE sol. = 0.200
Exact sol. 0.221 0.174 0.127 0.083 0.000 ODE sol. = 0.200

T = 0.400
App. sol. 0.493 0.379 0.273 0.176 0.002 ODE sol. = 0.400
Exact sol. 0.492 0.377 0.271 0.174 0.000 ODE sol. = 0.400

T = 0.800
App. sol. 1.229 0.901 0.622 0.384 0.008 ODE sol. = 0.798
Exact sol. 1.226 0.896 0.616 0.377 0.000 ODE sol. = 0.800

T = 1.600
App. sol. 3.959 2.610 1.629 0.917 0.027 ODE sol. = 1.594
Exact sol. 3.953 2.597 1.612 0.896 0.000 ODE sol. = 1.600

T = 3.200
App. sol. 23.468 11.973 5.885 2.665 0.074 ODE sol. = 3.184
Exact sol. 23.533 11.936 5.821 2.597 0.000 ODE sol. = 3.200

Number of integration steps in time = 36
Number of function evaluations = 498
Number of Jacobian evaluations = 17
Number of iterations = 117

Example Program
Parabolic PDE Coupled with ODE using Finite-differences and BDF

